United States Department of the Interior

Geological Survey

Two algorithms in BASIC for the analysis of variance on a desk-top computer

by

Norma Rait

Open-File Report 80-2015

This report is preliminary and has not been reviewed for U.S. Geological Survey editorial standards.

1980

CONTENTS

ILLUSTRATIONS
(Figures follow references cited)

Introduction

An algorithm in BASIC has been developed for the desk-top mini-computer HP9830A* for statistical studies of rock standards analyzed by quantitative direct-current arc emission spectrography. The homogeneity of these rock standards must be determined, so that the geochemist and chemist can use them as references.

Three bottles containing samples of the rock standards are chosen at random from those available. Two samples are taken from each bottle and are arced in random order. Their spectra are recorded on glass photoplates. The concentrations of the selected elements in each spectrum from each sample are determined (Bastron and others, 1960). These concentrations then are used for an analysis of variance in the tabular form of the BASIC algorithm.

Both algorithms reported in this communication (general and special tabular form) are being used to check new analytical methods for accuracy, by comparing the mean with recommended values, and also for precision.

* The mention of a specific product is for the purpose of identification only and does not constitute endorsement by the U. S. Geological Survey.

Equations

The equations for the algorithms were obtained from F. J. Flanagan (personal communication, 1978), Dixon and Massey (1957), and Box and others (1978).

X_{ij} = concentration of an element in the i^{th} bottle

T_i = total of all the measured concentrations of the same element in the i^{th} bottle

$T_i = X_{i1} + X_{i2} + \ldots\ldots + X_{ij}$

grand total = $G = T_1 + T_2 + T_3 + \ldots\ldots + T_i$

I = total number of bottles used

J = total number of determinations per bottle

mean = $M = G/(I \cdot J)$

correction factor = $C = G^2/(I \cdot J)$

The sums of squares (SS) are as follows:

1) SS bottle means = $B1 = (T_1^2 + T_2^2 + T_3^2 + \ldots\ldots + T_1^2)/J-C$

2) SS total variation = $T1 = (X^2_{11} + X^2_{12} + X^2_{21} + \ldots\ldots + X^2_{ij})-C$

3) SS within = $W1 = T1 - B1$

The degrees of freedom (DF) are as follows:

1) DF bottle means = $B2 = I-1$

2) DF within (Error) = $W2 = I$

3) DF total = $T2 = I \cdot J - I$

The mean sums of squares (MSS) are as follows:

1) MSS bottle = $B3 = B1/(I-1)$

2) MSS within = $W3 = W1/I$

3) MSS total = $T3 = T1/(I \cdot J-1)$

F ratio = B1/W1. When the F ratio is less than that in F tables (Box and others, 1978) at a given percentile level and for appropriate degrees of freedom, then the bottle variance is not significant; therefore, one may infer that the sample is homogeneous.

difference = D = B3 - W3 (can be negative)

bottle variance = E = D/J

bottle standard deviation = E1 = $E^{1/2}$

error standard deviation = H = $(W3)^{1/2}$

The above equations are used to write two BASIC algorithms. The first is the "Analysis of Variance for a one-way design of I bottles of a sample, J determinations per bottle, and Z elements". The second is a special case of the first algorithm, and it is in tabular form. This algorithm is the "Analysis of Variance for a one-way design of three bottles of a sample and two determinations per bottle". These algorithms are described below; both algorithms may be modified to meet the needs of the user.

General Statistical Analysis of Variance Algorithm

A detailed flow chart for the BASIC general statistical analysis of variance is given in figure 1 (Bycer, 1975). The general algorithm itself is in figure 2. An example of the printout is shown in figure 3. The date, job number, requestor's name, analyst's initials, photographic plate number, and title are on the first five lines. Printed on the next line is "Analysis of variance for a one-way design, I bottles of a sample with J determinations per bottle and Z elements". I, J, and Z are replaced by the number of bottles, number of determinations per bottle, and the number of elements, respectively, and these numbers in the example shown in figure 3 are 3, 2, and 1. Figure 3 illustrates the analysis of variance for determinations of Mg concentrations by a selected spectrographic method; the wavelength used for Mg is given in line 8. A table is printed listing the bottle number, the determination number for a given bottle, and the corresponding concentration in either percent or parts per million. The mean concentration is calculated and printed. Under this is another table containing five columns: the source of variation, sum of squares, degree of freedom, mean sum of squares, and F ratio. The bottle variance, the bottle standard deviation, and the error standard deviation are printed under the table. If the bottle variance is negative, "Bottle variance is negative" is printed. Then the error standard deviation is printed, but not the bottle standard deviation, which is an imaginary number. To simplify this algorithm, all data must be manually corrected for significant digits for the specific analytical method used.

Any part of this general BASIC algorithm may be altered. Photographic plates, time, laboratories, etc. may be used as the variable of classification instead of bottles.

Tabular Form of the Statistical Analysis of Variance Algorithm

The second algorithm is a special case of the first. It is the "Analysis of variance for a one-way design, three bottles of a sample and two determinations per bottle" in tabular form (figures 4 and 5).

Figure 6 shows the form of this algorithm that contains most information found in the general algorithm.

The tabular algorithm has an added feature not in the general program; after the concentrations are entered, they are all displayed. The user can check these entries for errors. If any entries are incorrect, all concentrations are reentered and checked again; thus, errors can be eliminated before calculations. Although this feature has satisfied the needs of this laboratory, it can be easily modified to check and correct individual entries. All data for quantitative DC arc spectroscopy should be corrected to two significant figures. Below the table, elements looked for by not found in quantitative emission spectrography are listed.

This program can also be modified to use other variables of classification (figure 7); thus, one may use two or more photographic plates, one bottle per standard, and two or more samples per standard per plate. The samples are arced in random order.

Summary

Neither algorithm corrects for significant figures, and data should be rounded for future use. Both algorithms may be modified to change the variable of classification; both algorithms may be used with any computer involving BASIC by making only minor adjustments in the program.

References Cited

Bastron, Harry, Barrett, P. R., and Murata, K. J., 1960, Method for the quantitative spectrochemical analysis of rocks, minerals, ores, and other materials by a powder D-C arc technique, U.S. Geological Survey Bulletin 1084-G, pp 165-182.

Box, G. E. P., Hunter, W. J., and Hunter, J. S., 1978, Statistics for experimenters, John Wiley & Sons, New York, 653 pp.

Bycer, B. B., 1975, Flowcharting: Programming, software designing, and computer problem solving, John Wiley & Sons, New York, 272 pp.

Dixon, W. J. and Massey, F. J., Jr., 1957, Introduction to statistical analysis, McGraw-Hill Book Co., Inc., New York, 488 pp.



Figure 1 - Flow chart for the BASIC algorithm for analysis of variance of a one-way design for I bottles of a sample and J determinations per bottle.



fig. 1 - (continued)

```
10 REM STATISTICAL STUDY PROGRAM (GENERAL)   WRITTEN BY N.RAIT
20 DIM L$[3],H$[3],Q$[20],R$[5],X$[15],S$[10],T$[79],T[100],X[100],P$[75],A$[3]
25 DIM W$[10]
30 DISP "DATE";
40 INPUT X$
50 DISP "YOUR INITIALS";
60 INPUT R$
70 DISP "FOR WHOM";
80 INPUT Q$
90 DISP "JOB #";
100 INPUT S$
110 DISP "PLATE #";
120 INPUT P$
130 PRINT "####################################################################"
140 PRINT
150 PRINT
160 PRINT X$,S$
170 WRITE (15,180)Q$
180 FORMAT "REQUESTOR : ",B
190 WRITE (15,200)R$
200 FORMAT "ANALYST :",B
210 PRINT
220 PRINT P$
230 PRINT
240 DISP "TITLE";
250 INPUT T$
260 PRINT T$
270 PRINT
280 PRINT
290 DISP "# BOTTLES,RUNS";
300 INPUT I,J
310 DISP "NUMBER OF ELEMENTS";
```

Figure 2 - General statistical analysis of variance algorithm

U. S. GOVERNMENT PRINTING OFFICE 1972 O - 477-066

```
320 INPUT Z
330 PRINT "ANALYSIS OF VARIANCE FOR A ONE-WAY DESIGN,"I"BOTTLES OF A"
340 PRINT "SAMPLE WITH"J"DETERMINATIONS PER BOTTLE, AND"Z"ELEMENTS"
350 PRINT
360 PRINT
370 PRINT
380 DISP "ELEMENT";
390 INPUT L$
400 DISP "WAVELENGTH";
410 INPUT W$
420 DISP "PPM";
430 INPUT A$
440 PRINT "ANALYSIS OF VARIANCE FOR "L$,W$
450 PRINT "****************************"
460 PRINT
470 G=C=M=S=Y=F=0
480 PRINT "BOTTLE  DETERMIN-  ANSWER"
490 IF A$="NO" THEN 530
500 IF A$="N " THEN 530
510 PRINT "  #     ATION #     PPM"
520 GOTO 540
530 PRINT "  #     ATION #      % "
540 FOR N=1 TO I
550 PRINT
560 LET T[N]=0
570 FOR O=1 TO J
580 DISP "BOTTLE"N"RESULT #"O;
590 INPUT X[O]
600 WRITE (15,610)N,O,X[O];
610 FORMAT 2X,F3.0,6X,F3.0,X,F10.1
620 T[N]=T[N]+X[O]
630 S1=(X[O])↑2
```

fig. 2 - (continued)

U. S. GOVERNMENT PRINTING OFFICE : 1959 O - 511171
867-106

```
640 REM S = SUM OF SQUARES OF DETERMINATIONS
650 S=S+S1
660 PRINT
670 NEXT O
680 REM Y=SUM OF THE TOTALS SQUARED PER BOTTLE
690 Y=Y+(T[N])↑2
700 REM G=GRAND TOTAL
710 G=G+T[N]
720 NEXT N
730 PRINT
740 REM M=MEAN
750 M=G/(J*I)
760 IF M<0.1 THEN 800
770 WRITE (15,780)M
780 FORMAT "MEAN =",F10.1
790 GOTO 830
800 WRITE (15,810)M
810 FORMAT "MEAN =",E10.1
820 REM C=CORRECTION FACTOR
830 PRINT
840 C=(G↑2)/(J*I)
850 REM T1=SUM OF SQUARES(TOTAL VARIATION)
860 T1=S-C
870 REM B1=SUM OF SQUARES(BOTTLE MEANS)
880 B1=Y/J-C
890 REM W1=SUM OF SQUARES WITHIN VARIATION
900 W1=T1-B1
910 REM B2=DEGREES OF FREEDOM BOTTLE MEANS
920 B2=I-1
930 REM W2=DEGREES OF FREEDOM WITHIN MEANS
940 W2=I*(J-1)
950 REM T2=DEGREES OF FREEDOM TOTAL
```

fig. 2 (continued)

U. S. GOVERNMENT PRINTING OFFICE : 1959 O - 511171
867-10C

```
960 T2=J*I-1
970 REM B3=MSS(BOTTLE MEANS)
980 B3=B1/B2
990 REM W3=MSS(WITHIN)
1000 W3=W1/W2
1010 REM F= RATIO OF BOTTLE MEANS TO WITHIN
1020 F=B3/W3
1030 PRINT "SOURCE OF          SUM OF     DEG. OF      MEAN SUM      F"
1040 PRINT "VARIATION          SQUARES    FREEDOM    OF SQUARES    RATIO"
1050 PRINT "------------------------------------------------------------"
1060 IF M<1 THEN 1170
1070 WRITE (15,1080)B1,B2,B3
1080 FORMAT "BOTTLE",/,"MEANS ",8X,F10.1,4X,F3.0,3X,F10.1
1090 WRITE (15,1100)F
1100 FORMAT 45X,F10.2
1110 WRITE (15,1120)W1,W2,W3
1120 FORMAT "ERROR",9X,F10.1,4X,F3.0,3X,F10.1
1130 PRINT "------------------------------------------------------"
1140 WRITE (15,1150)T1,T2
1150 FORMAT "TOTAL",9X,F10.1,4X,F3.0
1160 GOTO 1260
1170 WRITE (15,1180)B1,B2,B3
1180 FORMAT "BOTTLE",/;"MEANS ",8X,E10.1,4X,F3.0,4X,E10.1
1190 WRITE (15,1200)F
1200 FORMAT 45X,F10.2
1210 WRITE (15,1220)W1,W2,W3
1220 FORMAT "ERROR",9X,E10.1,4X,F3.0,4X,E10.1
1230 PRINT "------------------------------------------------------"
1240 WRITE (15,1250)T1,T2
1250 FORMAT "TOTAL",9X,E10.1,4X,F3.0
1260 PRINT
1270 PRINT
```

fig. 2 - (continued)

U. S. GOVERNMENT PRINTING OFFICE: 1959 O - 511171
887-100

```
1280 REM CALCULATIONS FOR BOTTLE STANDARD DEVIATIONS
1290 REM D=DIFFERENCE
1300 D=B3-W3
1310 IF D<0 THEN 1470
1320 REM E=BOTTLE VARIANCE
1330 E=D/J
1340 REM E1=BOTTLE STANDARD DEVIATION
1350 E1=E↑(1/2)
1360 IF D<0.1 THEN 1420
1370 WRITE (15,1380)E
1380 FORMAT "BOTTLE VARIANCE =",F10.2
1390 WRITE (15,1400)E1
1400 FORMAT "BOTTLE STANDARD DEVIATION =",F10.2
1410 GOTO 1480
1420 WRITE (15,1430)E
1430 FORMAT "BOTTLE VARIANCE =",E10.2
1440 WRITE (15,1450)E1
1450 FORMAT "BOTTLE STANDARD DEVIATION =",E10.2
1460 GOTO 1480
1470 PRINT "BOTTLE VARIANCE NEGATIVE"
1480 H=(W3)↑(1/2)
1490 IF H<0.1 THEN 1530
1500 WRITE (15,1510)H
1510 FORMAT "ERROR STANDARD DEVIATION =",F8.2
1520 GOTO 1550
1530 WRITE (15,1540)H
1540 FORMAT "ERROR STANDARD DEVIATION =",E10.2
1550 DISP "ANY MORE ELEMENTS";
1560 INPUT H$
```

fig. 2 - (continued)

U. S. GOVERNMENT PRINTING OFFICE : 1959 O - 511171
867-10C

```
1570 IF H$="YES" THEN 350
1580 IF H$="Y" THEN 350
1590 PRINT
1600 PRINT "######################################################### ########"
1610 PRINT
1620 PRINT
1630 STOP
1640 END
```

fig. 2 - (continued)

U. S. GOVERNMENT PRINTING OFFICE : 1959 O - 511171
867-10C

```
#####################################################################

1/31/79          AM95
REQUESTOR : FLANAGAN
ANALYST :NR

D242

QUANTITATIVE SPECTROGRAPHIC ANALYSIS, STANDARD RJM-22

ANALYSIS OF VARIANCE FOR A ONE-WAY DESIGN, 3    BOTTLES OF A
SAMPLE WITH 2    DETERMINATIONS PER BOTTLE, AND 1    ELEMENTS

ANALYSIS OF VARIANCE FOR MG   285.21NM
******************************

BOTTLE  DETERMIN-   ANSWER
  #     ATION #       PPM

   1       1        320.0
   1       2        240.0

   2       1        300.0
   2       2        340.0

   3       1        410.0
   3       2        450.0

MEAN =      343.3

SOURCE OF          SUM OF     DEG. OF      MEAN SUM        F
VARIATION          SQUARES    FREEDOM    OF SQUARES      RATIO
---------------------------------------------------------------
BOTTLE
MEANS.             24133.3       2         12066.7
                                                          7.54
ERROR               4800.0       3          1600.0
-------------------------------------------------------
TOTAL              28933.3       5

BOTTLE VARIANCE =    5233.33
BOTTLE STANDARD DEVIATION =      72.34
ERROR STANDARD DEVIATION =    40.00

#####################################################################
```

Figure 3 - An example of a printout for the general statistical analysis of variance algorithm

U. S. GOVERNMENT PRINTING OFFICE : 1959 O - 511171
867-100



Figure 4 - Flow chart for the BASIC algorithm for analysis of variance of a one-way design of three bottles of a sample and two determinations per bottle.


Since divisions can not be made by zero, W1 needs to be checked. (W1 is ≥ 0)
W1 = ∅?
Yes
No
B3 = B1/2
W3 = W1/3
H = (W3)↑(½)
F = B3/W3
D = B3-W3
MSS Bottle means
MSS Within
Error standard deviation
Ratio of bottle means to within
Difference
Check entries next
X(1,1); X(1,2); X(2,1)
Check 1st 3 ans.
X(2,2); X(3,1); X(3,2)
Check last 3 ans.
any errors?
Yes of Y or No or N
Errors?
Yes or Y
No or N
W1 = ∅?
ppm?
No or N
D < ∅ ?
E = D/2
E1 = E↑(½)
Bottle variance
Bottle stand. dev.
write
Print
disp
Elements not found?
F$
Elements looked for but not found: F$
END
any more elements?
Yes or Y
No or N

fig. 4 cont.

```
10 REM STATISTICAL STUDY PROGRAM (TABULAR)          WRITTEN BY N.RAIT
20 PRINT "INSTRUCTIONS - IF DETERMINATION IS LESS THEN ONE PERCENT "
30 PRINT "CHANGE TO PARTS PER MILLION BEFORE DOING CALCULATIONS."
40 PRINT "IF BY ACCIDENT YOU SAY NO WHEN ASKED ANY MORE ELEMENTS"
50 PRINT "THEN PRESS....END,RUN620,EXECUTE"
60 PRINT
70 PRINT
80 PRINT
90 DIM L$[3],H$[3],Q$[20],R$[5],X$[15],S$[10],T$[72],P$[3],T[10],X[3,2]
100 DIM B$[62],E$[3],O$[75],M$[50],J$[80],D$[80],F$[72]
110 DISP "DATE";
120 INPUT X$
130 DISP "YOUR INITIALS";
140 INPUT R$
150 DISP "FOR WHOM";
160 INPUT Q$
170 DISP "JOB #";
180 INPUT S$
190 DISP "PLATE NO.";
200 INPUT B$
210 LET D$="########################################"
220 D$[41,80]=D$
230 WRITE (15,240)D$
240 FORMAT B
250 PRINT
260 PRINT
270 PRINT X$,Q$,R$,S$
280 PRINT
290 PRINT "PLATE NO. "B$
300 PRINT
310 PRINT
320 DISP "TITLE";
```

Figure 5 - Tabular statistical analysis of variance algorithm

U.S. GOVERNMENT PRINTING OFFICE 1972 O - 477-064

```
330 INPUT T$
340 PRINT T$
350 PRINT
360 PRINT
370 PRINT "ANALYSIS OF VARIANCE FOR A ONE-WAY DESIGN,THREE BOTTLES OF A"
380 PRINT "SAMPLE WITH TWO DETERMINATIONS PER BOTTLE."
390 PRINT
400 PRINT
410 PRINT
420 PRINT "                    BOTTLES                        STANDARD DEVIATION"
430 LET M$="   MEAN     "
440 M$[12,22]="BOTTLES    "
450 M$[23,34]="ERROR      "
460 M$[35,43]="F        "
480 LET J$="----------------------------------------"
490 J$[41,69]=J$
500 WRITE (15,510)M$
510 FORMAT "ELE.",26X,B
520 LET O$="1          2          3                    "
530 O$[34,43]="2DF       "
540 O$[44,51]="3DF     "
550 O$[52,62]=" RATIO     "
570 WRITE (15,580)O$
580 FORMAT 10X,B
590 WRITE (15,600)J$[1,80]
600 FORMAT B
610 PRINT
620 DISP "ELEMENT";
630 INPUT L$
640 DISP "PPM";
```

fig. 5 - (continued)

U. S. GOVERNMENT PRINTING OFFICE : 1959 O - 511171
887-100
1287

```
650 INPUT P$
660 G=C=M=S=Y=F=0
670 FOR N=1 TO 3
680 LET T[N]=0
690 FOR O=1 TO 2
700 DISP "BOTTLE"N"RESULT #"O;
710 INPUT X[N,O]
720 T[N]=T[N]+X[N,O]
730 S1=(X[N,O])↑2
740 S=S+S1
750 NEXT O
760 Y=Y+(T[N])↑2
770 G=G+T[N]
780 NEXT N
790 M=G/6
800 C=(G↑2)/6
810 T1=S-C
820 B1=Y/2-C
830 W1=T1-B1
840 IF W1=0 THEN 900
850 B3=B1/2
860 W3=W1/3
870 H=(W3)↑(1/2)
880 F=B3/W3
890 D=B3-W3
900 DISP "CHECK ENTRIES NEXT"
910 WAIT 2000
920 DISP X[1,1];X[1,2];X[2,1]
930 WAIT 6000
940 DISP X[2,2];X[3,1];X[3,2]
950 WAIT 6000
960 DISP "ANY ERRORS";
```

fig. 5 - (continued)

U. S. GOVERNMENT PRINTING OFFICE : 1959 O - 511171
867-100

```
970 INPUT E$
980 IF E$="YES" THEN 620
990 IF E$="Y" THEN 620
1000 IF W1=0 THEN 1260
1010 IF D<0 THEN 1150
1020 E=D/2
1030 E1=E↑(1/2)
1040 IF P$="NO" THEN 1110
1050 IF P$="N" THEN 1110
1060 WRITE (15,1070)L$,X[1,1],X[2,1],X[3,1]
1070 FORMAT F10.0,F8.0,F8.0
1080 WRITE (15,1090)"PPM",X[1,2],X[2,2],X[3,2],M,E1,H,F
1090 FORMAT F9.0,F8.0,F8.0,F8.2,3X,F8.2,F10.2,2X,F8.2
1100 GOTO 1350
1110 WRITE (15,1120)L$,X[1,1],X[2,1],X[3,1]
1120 FORMAT F11.2,F8.2,F8.2
1130 WRITE (15,1135)" % ",X[1,2],X[2,2],X[3,2],M,E1,H,F
1135 FORMAT F10.2,F8.2,F8.2,F7.2,4X,F7.2,3X,F7.2,1X,F9.2,3X
1140 GOTO 1350
1150 IF P$="NO" THEN 1210
1160 IF P$="N" THEN 1210
1170 WRITE (15,1180)L$,X[1,1],X[2,1],X[3,1]
1180 FORMAT F10.0,F8.0,F8.0,F8.2,3X,F10.2,2X,F8.2,F10.2
1190 WRITE (15,1195)"PPM",X[1,2],X[2,2],X[3,2],M,"     NEG   ",H,F
1195 FORMAT F9.0,F8.0,F8.0,F8.2,3X,F8.2,F10.2
1200 GOTO 1350
1210 WRITE (15,1220)L$,X[1,1],X[2,1],X[3,1]
1220 FORMAT F11.2,F8.2,F8.2
1230 WRITE (15,1240)" % ",X[1,2],X[2,2],X[3,2],M,"     NEG    ",H,F
1240 FORMAT F10.2,F8.2,F8.2,F7.2,4X,F7.2,3X,F7.2
1250 GOTO 1350
1260 IF P$="YES" THEN 1320
```

fig. 5 - (continued)

U. S. GOVERNMENT PRINTING OFFICE · 1972 O - 457-084

```
1270 IF P$="Y" THEN 1320
1280 WRITE (15,1290)L$,X[1,1],X[2,1],X[3,1]
1290 FORMAT F11.2,F8.2,F8.2,F7.2
1300 WRITE (15,1240)" % ",X[1,2],X[2,2],X[3,2],M
1310 GOTO 1350
1320 WRITE (15,1330)L$,X[1,1],X[2,1],X[3,1]
1330 FORMAT F10.0,F8.0,F8.0,F8.0
1340 WRITE (15,1090)"PPM",X[1,2],X[2,2],X[3,2],M
1350 DISP "ANY MORE ELEMENTS";
1360 INPUT H$
1370 IF H$="YES" THEN 610
1380 IF H$="Y" THEN 610
1390 PRINT
1400 LET D$="########################################"
1410 D$[41,80]=D$
1420 WRITE (15,1430)D$[1,80]
1430 FORMAT B
1440 PRINT
1450 PRINT
1460 DISP "ELEMENTS NOT FOUND";
1470 INPUT F$
1480 PRINT "ELEMENTS LOOKED FOR BUT NOT FOUND:"
1490 PRINT F$
1500 PRINT
1510 PRINT
1520 STOP
1530 END
```

fig. 5 - (continued)

U. S. GOVERNMENT PRINTING OFFICE : 1959 O - 511171
867-100

```
################################################################################

1/31/79          FLANAGAN        NR               AM95

PLATE NO. D242,D260,D266,B258

QUANTITATIVE SPECTROGRAGPHIC ANALYSIS, STANDARD RJM-22

ANALYSIS OF VARIANCE FOR A ONE-WAY DESIGN,THREE BOTTLES OF A
SAMPLE WITH TWO DETERMINATIONS PER BOTTLE.
```

ELE.	BOTTLES 1	BOTTLES 2	BOTTLES 3	MEAN	STANDARD DEVIATION BOTTLES 2DF	STANDARD DEVIATION ERROR 3DF	F RATIO
FE	0.97	1.20	0.96				
%	1.20	0.93	1.10	1.06	NEG	0.16	0.06
MG	320	300	410				
PPM	240	340	450	343.33	72.34	40.00	7.54
CA	2400	260	2500				
PPM	2000	2100	2600	1976.67	458.62	769.81	1.71

```
################################################################################

ELEMENTS LOOKED FOR BUT NOT FOUND:
P<0.8%,NI<8PPM,CO<8,MO<8,NB<80,SC<8
```

Figure 6 - Example of a printout of the tabular algorithm

U. S. GOVERNMENT PRINTING OFFICE : 1959 O - 511171
867-100

```
##########################################################################

12/13/79        GAERTNER        NR                80%AR20%02

PLATE NO. G190,191,192

VARIATION BETWEEN PHOTOGRAPHIC PLATES FOR MN

ANALYSIS OF VARIANCE FOR A ONE-WAY DESIGN,THREE PHOTOGRAPHIC PLATES
WITH DUPLICATE SPECTRA PER STANDARD PER PLATE
```

STD.	PLATES 1	PLATES 2	PLATES 3	MEAN	STANDARD DEVIATION PLATES 2DF	STANDARD DEVIATION ERROR 3DF	F RATIO
G-2	360	212	287				
PPM	268	275	258	276.67	12.22	47.04	1.14
W-1	1307	942	1238				
PPM	1212	1086	877	1110.33	61.81	163.34	1.29

```
##########################################################################

COMMENTS:
NONE
```

Figure 7 - Example of a printout for a modified tabular algorithm